```
STUDENT > # Original system:  Ground===K1===M1.
STUDENT > # Modified system: Ground===K1==M1===K2===M2
STUDENT > #  where M2<<M1
STUDENT > # where both K1 and K2 have damping elements C1, C2 in
          parallel
STUDENT >
STUDENT > # Can we assume that the new mode introduced by addition
          of K2 M2 involves very little movement of M1?
STUDENT >
STUDENT > # Values to be
          studied:[M1=1000,M2=100,K1=1000,K2=10,C1=10,C2=0.1];
STUDENT > # Original M1/K1 system has resonant frequency
          w=sqrt(1/1)=1
STUDENT > # Added system creates a zero freq of M1 at
          w=sqrt(1/10)~0.3 (dynamic absorber effect)
STUDENT >
STUDENT > # Composite system has a root just above 1 (1.005) and a
          root below the zero at 0.3
STUDENT > # Question: Does the root at 0.3 affect M1?
STUDENT > # Answer:
STUDENT > # Figure 1 shows comparatively small peak in M1 response
          at 0.3 when exciting force applied at M1
STUDENT > # Figure 2 shows comparatively large peak in M1 response
          at 0.3 when exciting force applied at M2
STUDENT > # Figure 3 shows comparatively small peak in M1 response
          at 0.3 when base is moved
STUDENT > # Figure 4 is a zoom-in of Figure 3.
STUDENT > #  Results of course depend on magnitudes including
          damping
STUDENT >
STUDENT >
STUDENT > restart; # PART 1 - TRANSFER FUNCTION ANALYSIS (FORCES
          APPLIED TO MASSES)
STUDENT > with(inttrans):
STUDENT > Eq1:=F1applied(t)=m1*diff(x1(t),t,t)+k1*x1(t)+k2*(x1(t)-x2
          (t))+c1*diff(x1(t),t)+c2*(diff(x1(t),t)-diff(x2(t),t));
```

$$Eq1 := \mathrm{F1applied}(t) = m1\left(\frac{\partial^2}{\partial t^2}\mathrm{x1}(t)\right) + k1\,\mathrm{x1}(t) + k2\,(\mathrm{x1}(t) - \mathrm{x2}(t)) + c1\left(\frac{\partial}{\partial t}\mathrm{x1}(t)\right) + c2\left(\left(\frac{\partial}{\partial t}\mathrm{x1}(t)\right) - \left(\frac{\partial}{\partial t}\mathrm{x2}(t)\right)\right)$$

```
STUDENT > Eq2:=F2applied(t)=m2*diff(x2(t),t,t)+k2*(x2(t)-x1(t))+c2*(
          diff(x2(t),t)-diff(x1(t),t));
```

$$Eq2 := \mathrm{F2applied}(t) = m2\left(\frac{\partial^2}{\partial t^2}\mathrm{x2}(t)\right) + k2\,(\mathrm{x2}(t) - \mathrm{x1}(t)) + c2\left(\left(\frac{\partial}{\partial t}\mathrm{x2}(t)\right) - \left(\frac{\partial}{\partial t}\mathrm{x1}(t)\right)\right)$$

```
STUDENT > Eq1s:=laplace(Eq1,t,s):
STUDENT > subs1:=[laplace(F1applied(t),t,s)=F1(s),
          laplace(F2applied(t),t,s)=F2(s),
          laplace(x1(t),t,s)=X1(s),laplace(x2(t),t,s)=X2(s),
          x1(0)=0, x2(0)=0, D(x1)(0)=0, D(x2)(0)=0]:
STUDENT > Eq1s:=subs(subs1,Eq1s);
```

$$Eq1s := \mathrm{F1}(s) = m1\,s^2\,\mathrm{X1}(s) + k1\,\mathrm{X1}(s) + k2\,(\mathrm{X1}(s) - \mathrm{X2}(s)) + c1\,s\,\mathrm{X1}(s) + c2\,(s\,\mathrm{X1}(s) - s\,\mathrm{X2}(s))$$

```
STUDENT > Eq2s:=laplace(Eq2,t,s):
STUDENT > Eq2s:=subs(subs1,Eq2s);
```

$$Eq2s := \mathrm{F2}(s) = m2\,s^2\,\mathrm{X2}(s) + k2\,(\mathrm{X2}(s) - \mathrm{X1}(s)) + c2\,(s\,\mathrm{X2}(s) - s\,\mathrm{X1}(s))$$

```
STUDENT > solution:=solve({Eq1s,Eq2s},{X1(s),X2(s)}):
STUDENT > X1soln:=subs(solution,X1(s));
```

$$X1soln := (k2\,\mathrm{F1}(s) + c2\,s\,\mathrm{F1}(s) + m2\,s^2\,\mathrm{F1}(s) + \mathrm{F2}(s)\,k2 + \mathrm{F2}(s)\,c2\,s) \Big/ (k2\,m1\,s^2 + k2\,k1 + k2\,c1\,s + c2\,s^3\,m1 + c2\,s\,k1 + c2\,s^2\,c1 + m2\,s^4\,m1 + m2\,s^2\,k1 + m2\,s^2\,k2 + m2\,s^3\,c1 + m2\,s^3\,c2)$$

```
STUDENT > H11:=simplify(subs(F2(s)=0,X1soln)/F1(s)); # H11 =
          Transfer function of M1 motion in response to F1
```

$$H11 := (k2 + c2\,s + m2\,s^2) \Big/ (k2\,m1\,s^2 + k2\,k1 + k2\,c1\,s + c2\,s^3\,m1 + c2\,s\,k1 + c2\,s^2\,c1 + m2\,s^4\,m1 + m2\,s^2\,k1 + m2\,s^2\,k2 + m2\,s^3\,c1 + m2\,s^3\,c2)$$

```
STUDENT > H12:=simplify(subs(F1(s)=0,X1soln)/F2(s)); # H12 =
          Transfer function of M1 motion in response to F2
```

$$H12 := (k2 + c2\,s) / (k2\,m1\,s^2 + k2\,k1 + k2\,c1\,s + c2\,s^3\,m1 + c2\,s\,k1 + c2\,s^2\,c1 + m2\,s^4\,m1 + m2\,s^2\,k1 + m2\,s^2\,k2 + m2\,s^3\,c1 + m2\,s^3\,c2)$$

```
STUDENT >
STUDENT > system1:=[m1=1000,m2=100,k1=1000,k2=10,c1=10,c2=0.1]: #
          Define system parameters
STUDENT > H11mag:=evalc(abs(subs(s=I*w,subs(system1,H11)))); #
          magnitude of transfer fn
```

$$H11mag := \sqrt{\left(\frac{(10 - 100\,w^2)\,\%2}{\%2^2 + \%1^2} + .1\,\frac{w\,\%1}{\%2^2 + \%1^2}\right)^2 + \left(.1\,\frac{w\,\%2}{\%2^2 + \%1^2} - \frac{(10 - 100\,w^2)\,\%1}{\%2^2 + \%1^2}\right)^2}$$

$$\%1 := 200.0\,w - 1110.0\,w^3$$

$$\%2 := -111001.0\,w^2 + 10000 + 100000\,w^4$$

```
STUDENT > plot(H11mag,w=0.01..1.1,title='Fig1_Force_Applied_To_M1',n
          umpoints=10000);
```

Fig1_Force_Applied_To_M1




```
STUDENT > # zoom in on above reveals both pole and zero near 0.3
STUDENT > H12mag:=evalc(abs(subs(s=I*w,subs(system1,H12)))); #
          magnitude of transfer fn
```

$$H12mag := \sqrt{\left(10\,\frac{\%2}{\%2^2+\%1^2}+.1\,\frac{w\,\%1}{\%2^2+\%1^2}\right)^2+\left(.1\,\frac{w\,\%2}{\%2^2+\%1^2}-10\,\frac{\%1}{\%2^2+\%1^2}\right)^2}$$

$$\%1 := 200.0\,w - 1110.0\,w^3$$

$$\%2 := -111001.0\,w^2 + 10000 + 100000\,w^4$$

```
STUDENT > plot(H12mag,w=0.0001..1.1,title='Fig2_Force_Applied_To_M2'
          );
```

Fig2_Force_Applied_To_M2




```
STUDENT >
STUDENT >
STUDENT > restart; # PART 2 - TF ANALYSIS WITH BASE EXCITAITON
STUDENT > with(inttrans):
STUDENT > Eq1:=0=m1*diff(x1(t),t,t)+k1*(x1(t)-xb(t))+k2*(x1(t)-x2(t)
          )+c1*(diff(x1(t),t)-diff(xb(t),t))+c2*(diff(x1(t),t)-diff(
          x2(t),t));
```

$$Eq1 := 0 = m1\left(\frac{\partial^2}{\partial t^2}\mathrm{x1}(t)\right) + k1\,(\mathrm{x1}(t) - \mathrm{xb}(t)) + k2\,(\mathrm{x1}(t) - \mathrm{x2}(t))$$
$$+ c1\left(\left(\frac{\partial}{\partial t}\mathrm{x1}(t)\right) - \left(\frac{\partial}{\partial t}\mathrm{xb}(t)\right)\right) + c2\left(\left(\frac{\partial}{\partial t}\mathrm{x1}(t)\right) - \left(\frac{\partial}{\partial t}\mathrm{x2}(t)\right)\right)$$

```
STUDENT > Eq2:=0=m2*diff(x2(t),t,t)+k2*(x2(t)-x1(t))+c2*(diff(x2(t),
          t)-diff(x1(t),t));
```

$$Eq2 := 0 = m2\left(\frac{\partial^2}{\partial t^2}\mathrm{x2}(t)\right) + k2\,(\mathrm{x2}(t) - \mathrm{x1}(t)) + c2\left(\left(\frac{\partial}{\partial t}\mathrm{x2}(t)\right) - \left(\frac{\partial}{\partial t}\mathrm{x1}(t)\right)\right)$$

```
STUDENT > Eq1s:=laplace(Eq1,t,s):
STUDENT > subs1:=[laplace(xb(t),t,s)=XB(s),
          laplace(x1(t),t,s)=X1(s),laplace(x2(t),t,s)=X2(s),
          x1(0)=0, x2(0)=0, D(x1)(0)=0, D(x2)(0)=0, xb(0)=0,
          D(xb)(0)=0]:
STUDENT > Eq1s:=subs(subs1,Eq1s):
STUDENT > Eq2s:=laplace(Eq2,t,s);
```

$$Eq2s := 0 = m2\,(s\,(s\,\mathrm{laplace}(\mathrm{x2}(t), t, s) - \mathrm{x2}(0)) - \mathrm{D}(x2)(0))$$
$$+ k2\,(\mathrm{laplace}(\mathrm{x2}(t), t, s) - \mathrm{laplace}(\mathrm{x1}(t), t, s))$$
$$+ c2\,(s\,\mathrm{laplace}(\mathrm{x2}(t), t, s) - \mathrm{x2}(0) - s\,\mathrm{laplace}(\mathrm{x1}(t), t, s) + \mathrm{x1}(0))$$

```
STUDENT > Eq2s:=subs(subs1,Eq2s);
```

$$Eq2s := 0 = m2\,s^2\,\mathrm{X2}(s) + k2\,(\mathrm{X2}(s) - \mathrm{X1}(s)) + c2\,(s\,\mathrm{X2}(s) - s\,\mathrm{X1}(s))$$

```
STUDENT > solution:=solve({Eq1s,Eq2s},{X1(s),X2(s)}):
STUDENT > X1soln:=subs(solution,X1(s));
```

$$X1soln := \mathrm{XB}(s)\,(m2\,s^2\,k1 + m2\,s^3\,c1 + k2\,k1 + k2\,c1\,s + c2\,s\,k1 + c2\,s^2\,c1) \Big/ (m2\,s^4\,m1$$
$$+ m2\,s^2\,k1 + m2\,s^2\,k2 + m2\,s^3\,c1 + m2\,s^3\,c2 + k2\,m1\,s^2 + k2\,k1 + k2\,c1\,s + c2\,s^3\,m1 + c2\,s\,k1$$
$$+ c2\,s^2\,c1)$$

```
STUDENT > H:=simplify(X1soln/XB(s));
```

$$H := (m2\,s^2\,k1 + m2\,s^3\,c1 + k2\,k1 + k2\,c1\,s + c2\,s\,k1 + c2\,s^2\,c1) \Big/ (m2\,s^4\,m1 + m2\,s^2\,k1$$
$$+ m2\,s^2\,k2 + m2\,s^3\,c1 + m2\,s^3\,c2 + k2\,m1\,s^2 + k2\,k1 + k2\,c1\,s + c2\,s^3\,m1 + c2\,s\,k1 + c2\,s^2\,c1)$$

```
STUDENT >
STUDENT > system1:=[m1=1000,m2=100,k1=1000,k2=10,c1=10,c2=0.1]; #
          define system parameters
```

$$system1 := [m1 = 1000, m2 = 100, k1 = 1000, k2 = 10, c1 = 10, c2 = .1]$$

```
STUDENT > Hmag:=evalc(abs(subs(s=I*w,subs(system1,H)))); # magnitude
          of transfer fn
```

$$Hmag := \left(\left(\frac{(-100001.0\,w^2 + 10000)\,\%2}{\%2^2 + \%1^2} + \frac{(-1000\,w^3 + 200.0\,w)\,\%1}{\%2^2 + \%1^2}\right)^2\right.$$

$$+\left(\frac{(-1000\,w^3+200.0\,w)\,\%2}{\%2^2+\%1^2}-\frac{(-100001.0\,w^2+10000)\,\%1}{\%2^2+\%1^2}\right)^2\Bigg)^{1/2}$$

$$\%1 := -1110.0\,w^3+200.0\,w$$

$$\%2 := 100000\,w^4-111001.0\,w^2+10000$$

```
STUDENT > plot(Hmag,w=0.0001..1.1,title='Fig3_BASE_DISPLACEMENT',num
          points=10000);
```




```
STUDENT > plot(Hmag,w=0.2..0.4,title='Fig4_BASE_DISPLACEMENT_ZOOMED_
          IN',numpoints=10000);
```